

```
Blink | Energia 0101E0010
File Edit Sketch Tools Help
Blink $
/*
Blink
Egy másodpercre bekapcsoljuk a piros LED-et, azután egy
másodpercre lekapcsoljuk, s ezt ismételtetjük.


Ez a mintaprogram szabadon felhasználható (public domain).
*/

void setup() {
  // Digitális kimenetnek konfiguráljuk a piros LED-hez tartozó
  pinMode(RED_LED, OUTPUT);
}

void loop() {
  digitalWrite(RED_LED, HIGH); // bekapcsoljuk a LED-et
  delay(1000); // várunk egy másodpercig
  digitalWrite(RED_LED, LOW); // kikapcsoljuk a LED-et
  delay(1000); // várunk egy másodpercig
} |
19 LaunchPad w/ msp430g2553 (16MHz) on COM3
```



Energia





MSP430 programozás Energia környezetben Alfanumerikus LCD kijelzők



LCD = Liquid Crystal Display (folyadékkristály kijelző)

Folyadékkristály: olyan (szerves) anyag, mely sűrű folyadéknak tekinthető, ugyanakkor molekulái – a kristályokhoz hasonlóan – képesek struktúrákba rendeződni.




Az alapesetben csavart struktúra segít átjuttatni a polarizált fényt az elforgatott polárszűrőkön. Az elektromos tér átrendezi a molekulákat, már nem segítik a fény átjutását.


Forrás: http://en.wikipedia.org/wiki/Twisted_nematic_field_effect



LCD kijelző típusok



Grafikus pontmátrix kijelző



Szegmens kijelző




Alfanumerikus pontmátrix kijelző (4x20 karakter)



Alfanumerikus LCD

- HD44780 vagy kompatibilis vezérlő („ipari szabvány”)
- Háttérvilágítással ellátott, vagy reflexiós típus
- Csak rögzített karakterképet (betű, szám, írásjel) jeleníthetünk meg.
- 8x2, 16x2, 20x2, 20x4 az elterjedtebb forma



Sorszám	Jel	Funkció
1.	VSS	A tápegység negatív sarka, GND
2.	VCC	Tápfeszültség (többnyire +5 V)
3.	VEE	Kontraszt szabályozó jel (többnyire 0,5 V körül)
4.	RS	Parancs/adat választó (Register Select)
5.	R/W	Olvasás/írás (adatáramlási irány választása)
6.	E	Engedélyező jel (Enable)
7.	D0	Adatbusz 0.bit
8.	D1	Adatbusz 1.bit
9.	D2	Adatbusz 2.bit
10.	D3	Adatbusz 3.bit
11.	D4	Adatbusz 4.bit
12.	D5	Adatbusz 5.bit
13.	D6	Adatbusz 6.bit
14.	D7	Adatbusz 7.bit
15.	A	A háttérvilágítás pozitív sarka (LED anód)
16.	K	A háttérvilágítás negatív sarka (LED katód)



Lab11



HelloWorld – „Helló világ!” mintaprogram LCD kijelzővel



LCD_16x2 – A „HelloWorld” program bővített változata



LCD_16x2_voltmeter – Egyszerű feszültségmérő LCD kijelzéssel






LCD_thermometer – Analóg hőmérő LCD kijelzéssel



Hozzávalók

- 1 db LCD kijelző (16x2)
- 10 db átkötő vezeték (F + M)
- 1 db dugaszolós próbapanel
- 1 db Launchpad kártya

Háttérvilágítással ellátott kijelzőhöz további 2 db átkötő vezeték szükséges.





Bekötési vázlat

5 V-os kijelző meghajtása 4-bites módban

VCC = 5 V, VEE = 0 – 0,5V

RS → P2.0

R/W → GND (nem szabad olvasni!)

E → P2.1

D0 – (nem használt)

D1 – (nem használt)

D2 – (nem használt)

D3 – (nem használt)


D4 → P2.2

D5 → P2.3

D6 → P2.4

D7 → P2.5

A → +5V, K → GND (áramkorlátozás!)





Bekötési vázlat

3,3 V-os kijelző, 4-bites mód

VCC = 3.5 V, VEE = **nem használt!**

RS → P2.0

R/W → GND

E → P2.1

D0 – D3 (nem használt)

D4 → P2.2


D5 → P2.3

D6 → P2.4

D7 → P2.5

A → +3.5V, K → GND

(áramkorlátozás!)




Töltéspumpa:
csak a 3,3V-os
típusban van
beépítve!




Made with Fritzing.org




Optimális kontrasztbeállítás




1. Az általánosan javasolt megoldás



2. Fixen beállított kontraszt



3. Egyszerűsített kontraszt beállítás (néhány 5 V-os kijelzőnél bevált)



4. Egyszerűsített kontraszt beállítás 3,3V-os kijelzőhöz

Adatlap szerint 25 °C-on **VEE** számára **VCC – 4.5 V** körüli érték az optimális.



HelloWorld.ino (Helló világ)

```
#include "LiquidCrystal.h" // LCD "gyári" programkönyvtár becsatolása
LiquidCrystal lcd(P2_0,P2_1,P2_2,P2_3,P2_4,P2_5); //Példányosítás, láb kiosztás

void setup() {
  lcd.begin(16, 2); //Oszlopok és sorok száma
  lcd.print("hello, world!"); //Üzenet kiírása
}

void loop() {
  lcd.setCursor(0, 1); //Kurzor a második sor elejére
  lcd.print(millis()/1000); //Kiíratjuk a Reset óta eltelt időt
}
```

LiquidCrystal alkalmazói függvények

lcd.begin(oszlop,sor) – képernyő inicializálása

Lcd.setCursor(oszlop, sor) – kurzor beállítása a megadott helyre

Lcd.print("szöveg") – szöveg kiíratása

Lcd.print(kifejezés) – számérték kiíratása



HelloWorld.ino futtatása







LCD_16x2.ino

Egy kicsit megcifrázzuk az előző programot...

```
#include "LiquidCrystal.h"           // LCD "gyári" programkönyvtár becsatolása
LiquidCrystal lcd(P2_0,P2_1,P2_2,P2_3,P2_4,P2_5); //Példányosítás, lábkiosztás
int i, k;
void setup() {
  lcd.begin(16, 2);                 //Oszlopok és sorok száma
  lcd.print("<= 2 x 16 LCD =>");     //Kiírunk egy szöveget az első sorba
  // Látvány effekt: látszólag véletlen sorrendben írunk ki számjegyeket
  for(i=0; i<16; i++) {
    k = (i * 7) & 0x0F;
    lcd.setCursor(k, 1);           //Kiírás a második sorba
    lcd.print(k, HEX);            //Hexadecimális kiírás (0-9,A-F)
    delay(100);
  }
  delay(2000);
  lcd.clear();                     //Képernyő törlése
  lcd.print("Time since RESET");   //Kiírás az első sorba
}

void loop() {
  lcd.setCursor(0, 1);            //Kurzor a második sor elejére
  lcd.print(millis()/1000);       //Kiíratjuk a Reset óta eltelt időt
}
```






Egyszerű feszültségmérő

Megmérhetjük a valamelyik analóg bemenetre kapcsolt feszültség értékét és kijelezhetjük az LCD modul segítségével.

A rajzon egy potméter segítségével állítjuk be a feszültséget (*analóg vezérlésre is használható ez a módszer*) a **P1_4** lábón, de bármilyen egyenfeszültséget megmérhetünk így a 0 – 3,5 V tartományban.

A rajzon egy 5 V-os kijelző bekötése látható.



Made with  Fritzing.org



LCD_16x2_voltmeter.ino

```
#include "LiquidCrystal.h"           // LCD "gyári" programkönyvtár becsatolása
LiquidCrystal lcd(P2_0,P2_1,P2_2,P2_3,P2_4,P2_5); //Példányosítás, lábkiosztás

void setup() {
  lcd.begin(16, 2);                 //Oszlopok és sorok száma
  lcd.print("Analog Voltmeter");    //Kiírunk egy szöveget az első sorba
}

void loop() {
  int sensorValue = analogRead(A4); //Mérés az A4 bemeneten
  // Átszámítjuk a 0 - 1023 közötti eredményt voltra (0 - 3.5V):
  float voltage = sensorValue * (3.5 / 1023.0);
  lcd.setCursor(0, 1);              //Kurzor a második sor elejére
  lcd.print(voltage,3);              //Kiíratjuk az eredményt
  lcd.print(" V");
  delay(1000);
}
```

Megjegyzés: Alapértelmezetten VCC /GND lesz a Vref+/Vref- referencia feszültség




Analóg hőmérő

Egy analóg hőmérő jelét is megmérhetjük a beépített ADC segítségével. A fokokra átszámított hőmérsékletet kijelmezhetjük az LCD modul segítségével.

Az alábbi kapcsolásban a Lab08-ban már használt **MCP9700** hőmérő kimenetét kötjük a **P1_4** lábra.

A rajzon egy 5 V-os LCD kijelző bekötése látható.



Made with Fritzing.org



LCD_16x2_thermometer.ino

```
#include "LiquidCrystal.h"           // LCD "gyári" programkönyvtár becsatolása
LiquidCrystal lcd(P2_0,P2_1,P2_2,P2_3,P2_4,P2_5); //Példányosítás, láb kiosztás

void setup() {
  analogReference(INTERNAL1V5);      //Az 1,5 V-os belső referenciát választjuk
  lcd.begin(16, 2);                  //Oszlopok és sorok száma
  lcd.print("Thermometer");          //Kiírunk egy szöveget az első sorba
}

void loop() {
  long mysum = 0;                    //ebben összegezzük az eredményt
  for(int i=0; i<1500; i++) {
    mysum += analogRead(A4);
  }
  float voltage = mysum>>10;         //Osztas 1024-gyel
  float tempC = (voltage-500)/10;    //Átszámítás fokokra
  lcd.setCursor(0, 1);                //Kurzor a második sor elejére
  lcd.print(voltage,0);               //Kiíratjuk az eredményt
  lcd.print(" mV ");
  lcd.print(tempC,1);
  lcd.print(" C");
  delay(1000);
}
```